Pruning the SASLOG –
Digging into the Roots of NOTEs, WARNINGs, and ERRORs

Andrew T. Kuligowski
Nielsen Media Research

ABSTRACT

You've sat through constant design meetings. You've endured countless requests for "just one more little change". You even managed to find a creative solution to that nagging technical problem. But, you persevered, and despite all of the obstacles, you've managed to eliminate the final syntax error in your newest SAS routine. Time to sit back and relax -- uh, not quite ...

The primary focus of this presentation will be on techniques to ensure comprehension of your input data. We will look at several messages that are often found in the SASLOG, such as:

NOTE: MERGE statement has more than one data set with repeats of BY values.

that imply that there may be gaps in your knowledge of your data! Special emphasis will be placed on the use of ad-hoc queries to assist in finding data anomalies that can cause problems with your SAS code. It is assumed that the reader has a basic understanding of the SASLOG, including its composition, format, and the SAS system options which control its content.

MERGE Statement: Repeats of BY Values

Most users of the SAS system have encountered the following message:

NOTE: MERGE statement has more than one data set with repeats of BY values.

There are many papers in the Proceedings from past SUGI, SESUG, SCSUG, and the other various regional SAS User Group conferences that describe how to programatically force such a merge to occur. This assumes that the user wants to merge datasets that have repeats of BY values. However, it is also possible that the user did not expect this condition. This implies an error in your SAS routine, caused by a misunderstanding of the input data. We want to isolate these cases and modify our assumptions, so that we can correct our MERGE and eliminate this condition.

The following will illustrate an example of "repeats of BY values". We are going to merge a dataset containing a list of dog breeds [see Table "1-A"] against a dataset containing the dogs owned by sample households [see Table "1-B"] using the variable BREED, keeping only those records from the Breed file that have a corresponding record in the Household file. As one would expect (knowing, of course, that this example was created to illustrate the problem at hand), the merge results in "more than one data set with repeats of BY values" [illustrated in Table "1-C"]. The problem is to determine which BY values had repeats, which records (on which files) are affected, and what additional information needs to be included to make the "BY values", also referred to as "merge variables", unique on at least one of the files.

<table>
<thead>
<tr>
<th>BREED</th>
<th>VARIETY</th>
<th>OTHER INFO</th>
<th>BREED</th>
<th>VARIETY</th>
<th>OTHER INFO</th>
</tr>
</thead>
<tbody>
<tr>
<td>Bulldog</td>
<td></td>
<td>17834</td>
<td>Dachshund</td>
<td>Std Longhair</td>
<td>75324</td>
</tr>
<tr>
<td>Dalmatian</td>
<td></td>
<td>49235</td>
<td>Dachshund</td>
<td>Std Wirehair</td>
<td>09389</td>
</tr>
<tr>
<td>Dachshund</td>
<td>Mini</td>
<td>18435</td>
<td>Ger Sheprd</td>
<td></td>
<td>09622</td>
</tr>
<tr>
<td>Dachshund</td>
<td>MiniLonghair</td>
<td>75846</td>
<td>GoldRetrvr</td>
<td></td>
<td>38292</td>
</tr>
<tr>
<td>Dachshund</td>
<td>MiniWirehair</td>
<td>09431</td>
<td>Husky,Sib</td>
<td></td>
<td>75555</td>
</tr>
<tr>
<td>Dachshund</td>
<td>Std</td>
<td>18098</td>
<td>Lab Retrvr</td>
<td></td>
<td>38192</td>
</tr>
</tbody>
</table>

Table "1-A" : Breed File
We can use basic elements of the SAS System to do most of the analysis for us; our most powerful tool will be the MEANS procedure. PROC MEANS is traditionally used to compute descriptive statistics for numeric variables in a SAS Data set. In this situation, we simply need a list of the unique values of our merge variable (or, in a more complex case, the unique combinations of values for our merge variables), along with a count of the number of occurrences of each in both of our datasets. The NWAY option on the PROC MEANS statement will limit the output dataset to only those observations with the "highest interaction among CLASS variables" -- that is, only those records containing the unique values or combinations of values for the BY variables will be written to the output dataset. NOPRINT is optional; it can be used or excluded based on personal preference. The variable used in the VAR statement can be any numeric variable in the dataset; the only condition is that it must be a numeric variable. Date variables and ID values should be considered, since many / most datasets contain them and they are normally stored as numeric values. (In the rare event your dataset contains exclusively character variables, you will need to add a numeric variable to either the original dataset or a copy of it prior to issuing PROC MEANS.) The OUTPUT statement must specify an OUT= dataset. It must also include the statistics keyword N=, so that a record count -- and only a record count -- for each unique combination of values for the BY variables will be written to each observation of the output dataset. [Table "1-D" illustrates this use of PROC.]

The next step is to take the outputs of our PROC MEANS for each affected dataset and merge them together. We cannot encounter the "... repeats of BY values" note, since we now only have one observation per unique combination of BY values! Each observation does have a count of the number of observations that contain the combination of BY values in their original dataset, stored as _FREQ_ by default. The RENAME= option the datasets in the MERGE statement can be used to give the _FREQ_ variable in each dataset a unique name in our output dataset. (Alternatively, a cleaner approach would be to change the N= option on the OUTPUT statement in PROC MEANS to N=varname, avoiding the need for the subsequent RENAME.) We will use a subsetting IF, so that we only keep those observations that have multiple occurrences in each input dataset. The output of this step will contain the unique combinations of values that are causing the "...repeats of BY values" note. [The routine is contained in Table "1-E", and the output depicted in Table "1-F".]
Up to this point, we have not discussed one important factor in this analysis - the human element. The process described in this section is meant to be used as a tool to guide the analyst through the unknown elements in their data - once these areas become known - there is no need to continue this analysis. The listing of unique combinations of values that occur multiple times in each dataset will often be that stopping point for an analyst. The information obtained will allow them to make modifications to their assumptions and corresponding changes to their routines. However, in the event that the oddities are still not clear, we can employ one or more additional MERGE steps, taking the merged outputs from the PROC MEANS, and merging those dataset against each of the original datasets. [Table "1-G" shows how this is done, while Table "1-H" displays the results of this analysis.] This final step should provide sufficient clarification for the analyst to determine which factor(s) are missing in their assumptions and adjust their routines accordingly.
DATA CHKOWNED;
MERGE DOGOWNED (IN=IN_BREED)
     SUMMERGE (IN=IN_MERGE);
BY BREED;
IF IN_MERGE;
RUN;

NOTE: The data set WORK.CHKOWNED has 2 observations and
7 variables.

DATA CHKBREED;
MERGE DOGBREED (IN=IN_BREED)
     SUMMERGE (IN=IN_MERGE);
BY BREED;
IF IN_MERGE;
RUN;

NOTE: The data set WORK.CHKBREED has 6 observations and
5 variables.

Table "1-G" : SASLOG - Merging the analysis back to the original input

<table>
<thead>
<tr>
<th>OBS</th>
<th>HHL_ID</th>
<th>BREED</th>
<th>VARIETY</th>
<th>BIRTHDAY</th>
<th>GOTCHADY</th>
<th>CNTBREED</th>
<th>CNTOWNED</th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>5884</td>
<td>Dachshund</td>
<td>Std Longhair</td>
<td>12678</td>
<td>12870</td>
<td>6</td>
<td>2</td>
</tr>
<tr>
<td>2</td>
<td>5884</td>
<td>Dachshund</td>
<td>Std Longhair</td>
<td>13085</td>
<td>13179</td>
<td>6</td>
<td>2</td>
</tr>
</tbody>
</table>

Table "1-H" : Results of merging the analysis back to the original input

The end result of this analysis is the discovery that the BY statement in the routine does not contain the proper variables to uniquely identify each record. By adding the extra variable or variables to the original BY statement, the routine works without error – and without the offending NOTE. [Table "1-I" shows the corrected MERGE statement.]

DATA DOGDTAIL;
MERGE DOGOWNED (IN=IN_OWNED)
     DOGBREED (IN=IN_BREED);
BY BREED VARIETY;
IF IN_OWNED;
RUN;

NOTE: The data set WORK.DOGDTAIL has 9 observations and
6 variables.

Table "1-I" : SASLOG - MERGE without "repeats of BY values"
INPUT Statement: Reached past the end of a line

Most users of the SAS system have encountered the following message:

NOTE: SAS went to a new line when INPUT statement reached past
the end of a line.

The manuals describe how to prevent this message using options on the INFILE statement. MISSOVER and TRUNCOVER will prevent SAS from reading the next line but continue processing, while STOPOVER will force an error condition and stop building the data set. However, these options do not help resolve which line(s) on the input dataset triggered the problem. This message implies an error in your SAS routine, caused by a misunderstanding of the input data. We want to isolate these cases and modify our INPUT statement so that we no longer have this condition.

We can illustrate this with an example. We will read a sequential file containing the ID numbers of students in attendance at classes per day. Each record will have a key containing the Date and Class Name. The next two fields will represent the Number of Students Registered and the Number of Students Absent. Finally, the record will have a variable number of Student IDs, representing the students in attendance at the class on the given date [shown in Table “2-A”]. Our first attempt at reading this file will use the Number of Students Registered field to determine - incorrectly - how many occurrences of the Student ID field must be processed. [Table “2-B” will illustrate the SAS default of attempting to complete the INPUT statement on the next line. Note that the SASLOG indicates that we have 3 observations, although we read a file with 5 lines.] The problem is to isolate the incorrect assumption in our original analysis that is causing us to continue the INPUT after we have hit the end of line.

<table>
<thead>
<tr>
<th>DATE</th>
<th>CLASS</th>
<th>REGIST.</th>
<th>ABSENT</th>
<th>ID # OF STUDENTS IN ATTENDANCE</th>
</tr>
</thead>
<tbody>
<tr>
<td>02/21/96</td>
<td>Physics</td>
<td>12</td>
<td>2</td>
<td>27 29 33 34 37 41 42 43 44 45</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Botany</td>
<td>15</td>
<td>7</td>
<td>6    7    9    28 35 36 37 51</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Geology</td>
<td>16</td>
<td>9</td>
<td>13   29   30   31 39 45 46</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Anatomy</td>
<td>8</td>
<td>1</td>
<td>10   12   22   25 32 47 49</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Zoology</td>
<td>10</td>
<td>0</td>
<td>1    3    7    8   9  12 18 19 22 23</td>
</tr>
</tbody>
</table>

Table “2-A” : Attendance File

<table>
<thead>
<tr>
<th></th>
<th></th>
<th>DATA ATTEND;</th>
</tr>
</thead>
<tbody>
<tr>
<td>15</td>
<td></td>
<td>ARRAY ATNDID (25) ATNDID01-ATNDID25 ;</td>
</tr>
<tr>
<td>16</td>
<td></td>
<td>INFILE 'C:\SCSUG02\TSTS-SAS\ATTEND.DAT';</td>
</tr>
<tr>
<td>17</td>
<td></td>
<td>INPUT @ 1 DATE MMDDYY8.</td>
</tr>
<tr>
<td>18</td>
<td></td>
<td>@ 10 CLASS $CHAR8.</td>
</tr>
<tr>
<td>19</td>
<td></td>
<td>@ 18 REGIST 2.</td>
</tr>
<tr>
<td>20</td>
<td></td>
<td>@ 21 ABSENT 2. @ ;</td>
</tr>
<tr>
<td>21</td>
<td></td>
<td>pt = 24 ;</td>
</tr>
<tr>
<td>22</td>
<td></td>
<td>DO CNT = 1 TO REGIST;</td>
</tr>
<tr>
<td>23</td>
<td></td>
<td>INPUT @ pt ATNDID(CNT) 2. @ ;</td>
</tr>
<tr>
<td>24</td>
<td></td>
<td>pt = pt + 3 ;</td>
</tr>
<tr>
<td>25</td>
<td></td>
<td>END ;</td>
</tr>
<tr>
<td>26</td>
<td></td>
<td>RUN;</td>
</tr>
</tbody>
</table>

NOTE: The infile 'C:\SCSUG02\TSTS-SAS\ATTEND.DAT' is:
FILENAME=C:\SCSUG02\TSTS-SAS\ATTEND.DAT,
RECFM=V,LRECL=256

NOTE: 5 records were read from the infile 'C:\SCSUG02\TSTS-SAS\ATTEND.DAT'.
The minimum record length was 43.
The maximum record length was 52.
NOTE: SAS went to a new line when INPUT statement reached past the end of a line.
NOTE: The data set WORK.ATTEND has 3 observations and 31 variables.
NOTE: The DATA statement used 0.98 seconds.

Table “2-B” : Incorrect INPUT routine
As in the MERGE problem discussed earlier in this paper, we can use basic elements of the SAS System to do most of the analysis for us. In this case, we will use the options available on the INFILE statement itself. The LENGTH= option on the INFILE statement will define a numeric variable, which will be assigned the length of the current input line when an INPUT statement is executed. In our example, we will begin with the assumption that our initial input statement is correct and it is the array processing that is incorrect. Therefore, we will subtract the 22 bytes contained in our record's key from our line size. (Please note that the LENGTH= variable is not written to the output dataset; we will store its value in another SAS variable so that we can make further use of it in subsequent steps if necessary.) Finally, we will assume that our array is correctly made up of 2 character numeric variables, delimited by a single blank character. Therefore, we will divide the remaining line size by 3 to determine the number of members in the array per line; we will add 1 to the difference under the assumption that the final numeric value is not blank-padded. [Table "2-C" contains the SAS routine described in this paragraph, while Table "2-D" contains the dataset created by executing the routine.]

```
521 DATA LINELONG;
522   INFILE 'C:\SCSU\TSTS-SAS\ATTEND.DAT'
523      MISSOVER LENGTH=LNSZ;
524   INPUT @ 1 DATE MMDDYY8.
525         @ 10 CLASS $CHAR8.
526         @ 18 REGISTCT 2.
527         @ 21 ABSENTCT 2. @ ;
528   LINESIZE = LNSZ;
529   STDNTCNT = ( LINESIZE - 23 + 1 ) / 3;
530 RUN;
```

Table "2-C" : Using LENGTH= to determine line size

<table>
<thead>
<tr>
<th>DATE</th>
<th>CLASS</th>
<th>REGIST.</th>
<th>ABSENT</th>
<th>LINESIZE</th>
<th>STDNTCNT</th>
</tr>
</thead>
<tbody>
<tr>
<td>02/21/96</td>
<td>Physics</td>
<td>12</td>
<td>2</td>
<td>52</td>
<td>10</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Botany</td>
<td>15</td>
<td>7</td>
<td>46</td>
<td>8</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Geology</td>
<td>16</td>
<td>9</td>
<td>43</td>
<td>7</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Anatomy</td>
<td>8</td>
<td>1</td>
<td>46</td>
<td>8</td>
</tr>
<tr>
<td>02/21/96</td>
<td>Zoology</td>
<td>10</td>
<td>0</td>
<td>52</td>
<td>10</td>
</tr>
</tbody>
</table>

Table "2-D" : Results of our LENGTH= experiment

The final task is the trickiest one -- interpretation of the output. By quickly scanning the output, we can see that only one record has the same value, 10, for "REGIST" as for "STDNTCNT". The record has one other oddity -- the value of "ABSENT" is 0. As we all learned early on in life, 10 - 0 = 10. Therefore, we can adopt the working theory that STDNTCNT = REGIST - ABSENT. A quick check of our input data will show that the theory holds for every record in our dataset. (We could have written another ad-hoc routine to verify this assumption if our sample data had been more complex.) Therefore, by replacing the upper bound of our FOR loop with the calculated value STDNTCNT, our data can be read without error [as shown in Table "2-E"].

In the real world, the solution may not be as apparent as it was for our contrived example. In these cases, it may be necessary to review the assumptions made at the start of the analysis. The investigation may have cast a doubt on their validity, or possibly even disproved some of them altogether. The analyst should adjust the assumptions that are believed to be incorrect and return to exploring their data with a fresh angle. It is also possible that the investigation has not been fruitful, but none of the working assumptions have been neither proved nor disproved. The analyst has two choices at this point. One possibility is to return to the analysis using an alternate tool. [For example, during the research period for this paper, the author explored the use of the $VARYING. informat as a mechanism to solve the problem. It proved to be less effective than the LENGTH= option described above, and was omitted from the final draft in the interests of space.] The other approach is to return to the analyst's assumptions and alter one or more of them. The
analyst can then readdress the problem from a new angle -- the worst case will be that it proves no more fertile than the unproductive approach that the analyst had just abandoned!

```
DATA ATTEND;
ARRAY ATNDID (25) ATNDID01-ATNDID25 ;
INFILE 'C:\SCSUG02\TSTS-SAS\ATTEND.DAT';
INPUT @  1  DATE    MMDDYY8.
       @ 10 CLASS   $CHAR8.
       @ 18 REGIST   2.
       @ 21 ABSENT   2. @ ;
STDNTCNT = REGIST - ABSENT ;
pt = 24 ;
DO CNT = 1 TO STDNTCNT ;
   INPUT @ pt ATNDID(CNT) 2. @ ;
pt = pt + 3 ;
END ;
RUN;
```

Table "2-E" : Correct INPUT routine

**BEST. Format**

Most users of the SAS system have encountered the following message:

```
NOTE: At least one W.D format was too small for the number to be printed. The decimal may be shifted by the "BEST" format.
```

You can refer to the manuals to learn the use and benefits of the **BEST.w** format, which is the default for numeric variables. However, most users of the SAS System prefer to embellish their output by using the various output formats available to them. The message listed above informs you that SAS encountered a minor problem with their routine, and is overriding its original instructions in order to complete its task without error. The message implies, however, that you do not understand your data as well as you might. In order to eliminate this message, you will want to isolate the values that are too large for their format, and modify the formats on your PUT statements, so that you no longer have this condition.

In many cases, the offending data is blatantly obvious on the SAS routine's output. A quick visual scan of the report will identify the number or numbers whose formats have been adjusted for printing. You can simply correct those formats and re-execute your SAS routine, without having to perform an extensive analysis or needing to write and execute assorted ad-hoc routines. In other situations, the problematic output is not as easy to spot. For example, the report could be very large, with most of the values within it conforming to the expected format. Or, it might be in a .CSV format, which is harder for the typical human to scan over than columnar reports. In these cases, you can use some basic tools that the SAS system provides to probe your data.

Let us illustrate this situation with a very simple example -- let's say that you will input a series of numbers using a numeric 5. format, and attempt to output them with a numeric 4.2 format. [**Table "3-A"** displays the routine used to read and write the values. It also contains a table showing the actual value input by the program matched against the value output via the BEST. format.]
You need to use a basic assumption to validate the numbers in this example: a 4.2 format should produce a single-digit number, followed by a decimal point and two decimal places. Therefore, the decimal place should always be in the same position -- the second from the left -- when the number is printed. The PUT function can be used to store the number to a character variable using this selected format. (It is suggested that you use the Zw.d format, which zero-pads the number to the left if necessary. This will ensure that the assumption of aligned decimals will be valid in examples when the number in question is expected to be greater than 9.) Once the formatted value is stored electronically, you can use the INDEX function to determine if the decimal place is in the expected position. [Table "3-B" contains the validation ad-hoc, along with a tabular listing of its results.]

The combined use of the PUT and INDEX functions can also be used to isolate data that exceed the anticipated precision when your values do not contain decimal places. When a whole number exceeds the expected precision, the BEST. format override will use scientific notation. You can use the INDEX function to locate the first occurrence of "E" in your number, as formatted by the PUT function. A value of 0 indicates that "E" is not present; this is the expected condition. However, a non-zero value can be interpreted to mean that SAS converted our number to scientific notation.
This discussion of which values of the INDEX function are and are not valid under certain circumstances can get confusing. You may find it easier to convert them into text, to clearly differentiate valid from invalid values. [Table "3-C" contains a complete example of this validation technique, illustrating the search for both invalid decimal places and scientific notation, and the conversion of numeric responses into text values.]

```
673  DATA _NULL_;  
674    SET FORMAT42;  
675    C_FMT4_2 = PUT( FMT4_2, Z4.2 );  
676    WHERE_PT = INDEX( C_FMT4_2, '.' );  
677    WHERE_E  = INDEX( C_FMT4_2, 'E' );  
678    IF  WHERE_PT ^= 2  THEN  
679      ERRNOTE1 = 'DECIMAL';  
680    IF  WHERE_E ^= 0  THEN  
681      ERRNOTE2 = 'EXPONENTIAL';  
682    FILE LOG ;  
683    PUT @ 1  ACTUAL $CHAR5.  
684        @ 7  C_FMT4_2 $CHAR4.  
685        @ 13  ERRNOTE1 $CHAR10.  
686        @ 24  ERRNOTE2 $CHAR12.;  
687  RUN ;  
7.499 7.50  
14.49 14.5 DECIMAL  
768.1 0768 DECIMAL  
1997 1997 DECIMAL  
4858. 4858 DECIMAL  
54632 55E3 DECIMAL EXPONENTIAL  
NOTE: At least one W.D format was too small for the number to be printed. The decimal may be shifted by the "BEST" format.  
NOTE: The DATA statement used 0.28 seconds.  
```

Table "3-C" : Validation Ad-Hoc (and Results) #2

In this example, you can see that the largest value being read in contains 5 significant digits to the left of the decimal place. You can also see that the most precise value contains 3 decimal places. There are several possible solutions to our problem. If you want to print all values that you have read in, you can change the code to output the value using a Z9.3 format, or you can change the code to output the value using a Z8.2 format if you only need two decimal places. On the other hand, if you want to flag large numbers as erroneous values, you can put in validation code that watches for and traps numbers >=10, only printing values that can be properly displayed in the requested 4.2 format.

CONCLUSION

This paper addressed different messages that are commonly found in a SASLOG. It discussed the use of ad hoc routines to explore WHY those messages occurred, and covered how to correct a routine to prevent the recurrence of those messages. It is hoped that the mechanisms discussed in this paper might be used by the readers in their daily jobs. However, this paper is a failure -- at least in part -- if the process stops there. It is hoped, even more strongly, that the concepts of developing and using ad hoc routines to fully understand ones data are the true lessons that the reader retains from this paper.

REFERENCES / FOR FURTHER INFORMATION


SAS is a registered trademark or trademark of SAS Institute, Inc. in the USA and other countries. * indicates USA registration.

The author can be contacted via e-mail at:
A_Kuligowski@msn.com (preferred) or Andy.Kuligowski@NielsenMedia.com

ACKNOWLEDGMENTS

The author graciously offers thanks to Frank Di Iorio, Neil Howard, David Riba, Nancy Roberts, Tom Winn, Marje Fecht, and Debbie Buck for their contributions towards this paper. If they hadn't offered their suggestions on the content of this paper, agreed to proofread its myriad drafts, or offered the occasional reminder as to the definition of the word deadline, this paper would not have been completed. The author also wishes to acknowledge those individuals, whose identities have been lost to the ages, whose data oddities across the years provided the true inspiration behind this paper.